
Music Program for HP-28s, HP-48xx Calculator

Copyright © 1994 by Xah

1.What Does These Programs Do.

These programs can spell and play any triad or seventh chords. You type-in a chord and have the calculator to play it by pressing PLAY1, PLAY2, PLAY3 or spelled out by pressing SPELL. By pressing CHORD, HP-48 will generate a triad or 7th chord randomly. For example, you key in "A M " and press PLAY1, HP-48 will play a A major triad.

(Enharmonic chords with roots on E#, B#, Fb, Cb are not generated by CHORD but can be arguments of SPELL or PLAY. Enharmonic chords using double sharp or double flat as roots are not generated nor recognized. These programs can be used on HP-28s, HP-48s, HP-48sx, HP-48G, and HP-48GX models.

Chords to be PLAYed or SPELLed must be typed in as a string format:


"Note" + "AccidentalSign" + "Quality".

For example, A sharp major 7th is entered as "A#M7". A major triad is entered as "A M ". 

Here is a table showing all the chords starting with roots of A, Ab, and A#. Chords starting with other keys are similarly entered.

	
	A
	A#
	Ab

	
	"A M "
	"A#M "
	"AbM "

	
	"A m "
	"A#m "
	"Abm "

	Triad
	"A + "
	"A#+ "
	"Ab+ "

	
	"A ˚ "
	"A#˚ "
	"Ab˚ "

	
	"A M7"
	"A#M7"
	"AbM7"

	
	"A  7"
	"A# 7"
	"Ab 7"

	7th Chord
	"A m7"
	"A#m7"
	"Abm7"

	
	"A h7"
	"A#h7"
	"Abh7"

	
	"A ˚7"
	"A#˚7"
	"Ab˚7"


"M" = Major
"m" = minor
"+" = augmented
"h" = half-diminished.
"°" = diminished
    Spaces in the string can be left out.

This system of programs also allows you to key in the duration of each note to be played, in increment of 1/100th seconds. You can also change the ratio between adjacent notes to get 24 note per octave micro turning or other strange effects. To do so, simply change the value stored in DUR or R. For more information, see the variable diagrams in Section 3 of this book.

This system will be expanded in later version. The following functions are good candidates for expansion: play chord progressions, play scales, identify chords, invert/retrograde/augmentation/diminition of melodies, metronome, tuning machine. Readers are welcome to send in their .expansion programs.

2. Program Codes:

Here are the programs and their contents that you type into your calculator. The left side is the program name, right side is its content. Capital and lowercase letters are not interchangeable. Spaces must be exact; assume one space at the end of each line.

For convenience, SPELL CHORD SAME... PLAY1 PLAY2 PLAY3 SEMI DUR R should be typed in last so they appear in the first page of user menu. (build-in function ORDER in the Memory Menu can be used to re-arrange their order)

You may not want all these programs in your calculator.  For example, if you only use the program SPELL, you only need to type in SPELL and subprograms used by SPELL. (they are: UPD.M1 UPD.N1 UPD1 UPD2 UPD3 INTRV M.N CONV CRD1 INTR1 C D E F G A B) For a list of subprograms used by a program, check the required names box of that program diagram in Section Three of this book.

CHORD
« NOTE BCEF « RAND IF 0.4999 > THEN Q3RD ELSE Q7TH END » EVAL + »

SPELL
« UPD.M1 M.N »

PLAY1
« UPD.F1 LP1 »

PLAY2
« UPD.F1 LP2 »

PLAY3
« UPD.F1 LP3 »

SEMI
« 0 60 FOR s R s ^ 65.406 * DUR BEEP NEXT 0 60 FOR s 65.406391 32 * R s ^ / DUR BEEP NEXT »

R
1.05946309436

DUR
0.1

M.N
«  m1 « 1 4 FOR s m1 INTR1 s GET + CRD1 NUM CHR STR 2 GET s GET STR1 GET - IF DUP 3 ≥ THEN 12 - END CONV CRD1 NUM CHR STR 2 GET s GET SWAP + NEXT IF 12 FC? THEN DROP 3 LIST ELSE 4 LIST END » »

CONV
«  j « IF j 0 == THEN "" ELSE IF j 1 == THEN "#" ELSE IF j 2 == THEN "x" ELSE IF j -1 == THEN "b" ELSE IF j -2 == THEN "bb" ELSE IF j 3 == THEN "x#" ELSE IF j -3 == THEN "bbb" END END END END END END END » »

UPD.M1
« UPD.N1 1 GET IF 1 FS? THEN 1 + ELSE IF 2 FS? THEN 1 - END END »

UPD.N1
« UPD1 UPD2 UPD3 DUP NUM CHR STR »

UPD.F1
« UPD.N1 3 GET IF 1 FS? THEN R * ELSE IF 2 FS? THEN R / END END »

UPD1
« 'CRD1' STO 1 12 FOR f f CF NEXT IF CRD1 DUP "#" POS THEN 1 SF ELSE IF DUP "b" POS THEN 2 SF END END »

UPD2
« IF DUP "7" POS THEN 12 SF IF DUP "M" POS THEN 7 SF ELSE IF DUP "m" POS THEN 9 SF ELSE IF DUP "h" POS THEN 10 SF ELSE IF DUP "˚" POS THEN 11 SF ELSE 8 SF END END END END END »

UPD3
« IF 12 FC? THEN IF DUP "M" POS THEN 3 SF ELSE IF DUP "m" POS THEN 4 SF ELSE IF DUP "+" POS THEN 5 SF ELSE 6 SF END END END END INTRV »

INTRV
« IF 7 FS? THEN { 0 4 7 11 } ELSE IF 8 FS? THEN { 0 4 7 10 } ELSE IF 9 FS? THEN { 0 3 7 10 } ELSE IF 10 FS? THEN { 0 3 6 10 } ELSE IF 11 FS? THEN { 0 3 6 9 } ELSE IF 3 FS? THEN { 0 4 7 12 } ELSE IF 4 FS? THEN { 0 3 7 12 } ELSE IF 5 FS? THEN { 0 4 8 12 } ELSE IF 6 FS? THEN { 0 3 6 12 } END END END END END END END END END 'INTR1' STO »

LP1
«  f1 « 1 3 START IF 12 FS? THEN 5 ELSE 4 END 1 SWAP FOR t R INTR1 t GET ^ f1 * DUR BEEP NEXT NEXT » »

LP2
«  f1 « 0 5 FOR oc IF 12 FS? THEN 4 ELSE 3 END 1 SWAP FOR t R INTR1 t GET ^ 2 oc ^ * f1 * DUR BEEP NEXT NEXT » »

LP3
«  f1 « 5 0 FOR oc IF 12 FS? THEN 4 ELSE 3 END 1 FOR t R INTR1 t GET ^ 2 oc ^ * f1 * DUR BEEP -1 STEP -1 STEP » »

NOTE
« « { "A" "B" "C" "D" "E" "F" "G" } RAND 100 * FLOOR 7 MOD 1 + GET » EVAL « { "#" " " "b" } RAND 100 * FLOOR 3 MOD 1 + GET » EVAL + »


BCEF
«  { "B#" "E#" "Cb" "Fb" }  A B « 0 1 4 FOR j B j GET A SAME OR NEXT IF THEN A NUM CHR " " + ELSE A END » »

Q3RD
«  { "M" "m" "h" "°" } RAND 100 * FLOOR 4 MOD 1 + GET " " + »

Q7TH
« { "M" " " "m" "h" "°" } RAND 100 * FLOOR 5 MOD 1 + GET "7" + »

INTR1
0

CRD1
0

C
{ 36 { "C" "E" "G" "B" } 65.406 }

D
{ 38 { "D" "F" "A" "C" } 73.416 }

E
{ 40 { "E" "G" "B" "D" } 82.407 }

F
{ 41 { "F" "A" "C" "E" } 87.307 }

G
{ 43 { "G" "B" "D" "F" } 97.999 }

A
{ 45 { "A" "C" "E" "G" } 110 }

B
{ 47 { "B" "D" "F" "A" } 123.471 }

3. Program Diagrams and Descriptions

This section is for people who are interested to know how these program work. The most useful programs are CHORD, SPELL, PLAY, in which SPELL is the most difficult to program and CHORD is easily programed.

Here are some words with special meaning used in this section:

Names refer to the names of a user created object. It may be a program, a subprogram, a variable, or a constant.
  Letter name refer to the letter name of a note. For example, the Letter name for the note C# or Cb is C. On the other hand, Note means the note itself as opposed to letter name. For example, If we are speaking of D# major 7th chord, the letter names of the chord are D F A C, and the notes of the chord are D# Fx A# Cx.  To avoid confusion, it's important to pay attention to the word Letter name or Note in the following text.

Number are used to denote the twelve notes for computation. Thus, the notes C1 C#1 D1 D#1  E1 F1 F#1 G1 G#1 A1 A#1 B1 are represented by the number 1 2 3 4 . . . 12. But instead of using the number 1 to 12, we use the MIDI code 36 to 47. Tables are necessary for this conversion and for other tasks that is difficult or impossible to formulate.
The tables used can be seen in the seven constants and the program INTRV:

C
{ 36 { "C" "E" "G" "B" } 65.406 }

D
{ 38 { "D" "F" "A" "C" } 73.416 }

E
{ 40 { "E" "G" "B" "D" } 82.407 }

F
{ 41 { "F" "A" "C" "E" } 87.307 }

G
{ 43 { "G" "B" "D" "F" } 97.999 }

A
{ 45 { "A" "C" "E" "G" } 110 }

B
{ 47 { "B" "D" "F" "A" } 123.471 }

INTRV
« IF 7 FS? THEN { 0 4 7 11 } ELSE IF 8 FS? THEN { 0 4 7 10 } ELSE IF 9 FS? THEN { 0 3 7 10 } ELSE IF 10 FS? THEN { 0 3 6 10 } ELSE IF 11 FS? THEN { 0 3 6 9 } ELSE IF 3 FS? THEN { 0 4 7 12 } ELSE IF 4 FS? THEN { 0 3 7 12 } ELSE IF 5 FS? THEN { 0 4 8 12 } ELSE IF 6 FS? THEN { 0 3 6 12 } END END END END END END END END END 'INTR1' STO »

They include the MIDI code, seven possible letter names of a chord, and frequency for the seven white notes C1 to B1. The lists in INTRV are the interval sequence in semi-steps for different types of chords (for more detail, check the diagram of INTR1).

Here are the basic ideas behind the main programs:

CHORD is made by using random number to generate the seven letter names A to G, then again using random number to determine if accidental is to be added, what kind of chord (triad or 7th), and finally the quality. The kind of chord (triad or 7th) must be known before selecting quality because there are qualities that triad and 7th chord do not share (e.g. half-diminished).

SPELL: For example, the current chord is C#°7. First we convert the letter name of root (C) to MIDI code using the table we have: 36. Because the sharp sign, we add one to 36 to get the correct root of the chord: 37. With the help of the interval sequence table ({ 0 3 6 9 }in this case), we get the notes for C#°7 chord by adding 37 to each element: { 37 40 43 46 }. So far we have the correct notes of our chord. Now we check table again to get the correct letter names for our C# chord: { C E G B }. Convert this list to MIDI code we get: { 36 40 43 47 }. Now we have two lists, the correct notes of C#°7th chord { 37 40 43 46 }, and the correct letter names for that chord { 36 40 43 47 }. Subtract the first list from the second list element by element we get a new list:{ 1 0 0 -1 }. This new list tells us that the spelled chord will have a sharp for the first letter name and a flat for the fourth letter name. The final task is to get the letter names of our chord from the table once more: { C E G B }, then added the appropriate accidental signs, resulting the spelled C#°7: { C# E G Bb }.


In summary, spelling of a chord is possible by comparing every note in the chord with its correct letter name, then determine if accidentals are need, then add the accidentals to the correct letter names of that chord.

PLAY1 PLAY2 and PLAY3 use similar logics. Assume the current chord is C#°7. First, PLAY convert the letter name of root (C) to its frequency using the table we have, result: 65.406. Because the sharp sign, we multiply 65.406 by R (1.059..., the ratio between adjacent notes in well-tempered tuning) to get the correct frequency for the root of the chord: 69 (rounded). We now get the interval sequence for diminished 7th chord from the table in INTRV: { 0 3 6 9 }. With this sequence and root frequency (69) known, the frequencies for the chord is { 69R^0 69R^3 69R^6 69R^9 } or {69 82 98 117 }. The chord is now ready to be played using the built-in command BEEP. For PLAY1, the chord is repeated three times. In PLAY2, the chord is played five times and each time an octave higher. This is done by doubling the frequencies on each repeat. PLAY3 is like PLAY2 except it's from the highest note to the lowest.

Few other facts should be known before going further. In this book, Middle C is C3. The variable CRD1 is used to stored the current chord. The word current chrod and CRD1 are interchangeable.

Twelve flags are utilized in this program. (See the following table) Each flag mirrors a specific character of the current chord, such as it's quality. In this way, a specific character can be easily known by testing flags. Flag 1 & 2 register the accidental state of the root note. (flag 1 & 2 will not be both set) Flag 3 to 6 register the quality of the current chord  (for triads). Flag 7 to 11 register the quality of the current chord  (for 7th chord). Flag 12 indicate the current chord is either 7th or triad.
Flags & Interval Sequence Table

	Flag number
	Set means CRD1 is:
	Clear means:
	Interval sequence

	1
	#
	
	n/a

	2
	b
	
	n/a

	3
	M
	
	{ 0 4 7 12 }

	4
	m
	
	{ 0 3 7 12 }

	5
	+
	
	{ 0 4 8 12 }

	6
	˚
	
	{ 0 3 6 12 }

	7
	M7
	
	{ 0 4 7 11 }

	8
	7
	
	{ 0 4 7 10 }

	9
	m7 
	
	{ 0 3 7 10 }

	10
	h7
	
	{ 0 3 6 10 }

	11
	˚7
	
	{ 0 3 6 9 }

	12
	7TH Chord
	Triad
	n/a

	f1-2, accidental.

f3-6, quality for triads.

f5-11, quality for 7th chords.

f12, 7th or triad


Terms Used in Stack Diagrams:
	Term
	Possible value/Description

	AccidentalNumber
	-3,-2,-1,0,1,2,3

	MIDICode
	36,37,38,39,40,41,42,43,44,45,46,47
description: A number representing the notes from C1 to B1.

	"Note"
	"A#","A ","Ab","B#","B ","Bb","C#","C ","Cb". . . etc.

	"LetterName"
	"A ","B ","C ","D ","E ","F ","G "

	'LetterName'
	'A ','B ','C ','D ','E ','F ','G '

	 LetterName
	description: evaluated 'LetterName'
e.g. { 36 { "C" "E" "G" "B" } 65.406 } for 'C'.

	"AccidentalSign"
	"bbb","bb","b"," ","#","x","x#"

	"ChordQuality"
	"M ","m ","+ ","˚ ","M7"," 7","m7","h7","˚7"

	{Spelled Chord}
	description: a spelled chord in a list format.
e.g. {"E#" "G#" "B"}

	"Chord"
	"A M ","A m ","A + ","A ˚ ","A M7","A 7","A m7","A h7","A ˚7" . . . etc.
description: The above shows all the possible chords with root A. The spaces in the string have no significance. Argument of the type "Chord" can be typed without the spaces.


You will find a diagram for every name used in this music program. The diagram for the programs have five rows. The top section explain the naming of the program. "Logic" show the basic idea or method used for the program. "Stack take/give" is the stack diagram. "Result" section describes what the program do exactly. "Flags" show flags the program either checks or alters. "Names Required" is a list of names this program uses directly or indirectly. Each program is independent and functional providing the names in the Names Required list exist.

Program Diagrams/Descriptions

	SEMI
	play SEMI steps from c1 to c5
	

	Code
	« 0 60 FOR s R s ^ 65.406 * DUR BEEP NEXT 0 60 FOR s 65.406391 32 * R s ^ / DUR BEEP NEXT »
	

	Stack
take/give
	none
	none

	Result
	Play chromatic scale from c1 to c5 and back to c1.
	

	Names
required
	R DUR
	


	SPELL
	SPELL a chord
	

	Code
	« UPD.M1 M.N »
	

	Logic
	SPELL stores stack one in CRD1 then use it to returns a corrosponding MIDI code (m1) for root of that chord. m1 is then fed to M.N. M.N completes the spelling process
	

	Stack
take/give
	"Chord1"
	{Spelled Chord1}

	Result
	Uses argument to update CRD1, INTR1, flags 1-12, then returns a spelled current chord. e.g.: {"E#" "G#" "B"}.
	

	Flags
	1 2 3 4 5 6 7 8 9 10 11 12
	

	Names
required
	UPD.M1, UPD.N1, UPD1, UPD2, UPD3 INTRV, M.N, CONV, CRD1, INTR1 C D E F G A B
	


	M.N
	MIDI code versus MIDI code of the Note name
	

	Code
	«  m1 « 1 4 FOR s m1 INTR1 s GET + CRD1 NUM CHR STR 2 GET s GET STR1 GET - IF DUP 3 ≥ THEN 12 - END CONV CRD1 NUM CHR STR 2 GET s GET SWAP + NEXT IF 12 FC? THEN DROP 3 LIST ELSE 4 LIST END » »
	

	Logic
	Checks the current chord in CRD1 and compare the MIDI code for each of its note with the MIDI code for each of its letter name, thus calculate if accidentals should be placed after the letter name. M.N uses -3 -2 -1 0 1 2 3 to represent "bbb" "bb" "b" " " "#" "x" "x#". The conversion of numbers to accidental signs are done by the subprogram CONV.
	

	Stack
take/give
	MIDICode (of the root)
	{Spelled Chord1}

	Result
	Takes the MIDI code of the root of the current chord, returns the spelled current chord, i.e.: { "A" "C#" "E" "G#" }, for A major 7th chord.
	

	Flags
	none
	

	Names
required
	CONV  C D E F G A B
	


	CONV
	CONVert number to accidental signs
	

	Code
	«  j « IF j 0 == THEN "" ELSE IF j 1 == THEN "#" ELSE IF j 2 == THEN "x" ELSE IF j -1 == THEN "b" ELSE IF j -2 == THEN "bb" ELSE IF j 3 == THEN "x#" ELSE IF j -3 == THEN "bbb" END END END END END END END » »
	

	Stack
take/give
	AccidentalNumber
	"AccidentalSign"

	Result
	Converts -3 -2 -1 0 1 2 3 to "bbb" "bb" "b" " " "#" "x" "x#" symbols respectively.
	


	UPD.M1
	UPDate and return the MIDI code of root note
	

	
	« UPD.N1 1 GET IF 1 FS? THEN 1 + ELSE IF 2 FS? THEN 1 - END END »
	

	Stack
take/give
	"Chord1"
	MIDICode

	Result
	Uses argument to creat/update CRD1, INTRV, set flags 1-12 right, and return the MIDI code for the root note of the chord.
	

	Flags
	1 2 3 4 5 6 7 8 9 10 11 12
	

	Names
required
	UPD.N1 UPD1 UPD2 UPD3 INTRV CRD1 INTR1 C D E F G A B
	


	UPD.N1
	UPDate and return the Naturalized root note's table
	

	Code
	« UPD1 UPD2 UPD3 DUP NUM CHR STR »
	

	Stack
take/give
	1: "Chord1"
	1: LetterName

	Result
	Uses argument to creat/update CRD1, INTR1, flags 1 to 12, and return the naturalized root note of the chord. e.g.: returns { 36 { "C" "E" "G" "B" } 65.406 } for chords of the root C#, C, or Cb. 
	

	Flags
	1 2 3 4 5 6 7 8 9 10 11 12
	

	Names
required
	UPD1 UPD2 UPD3 INTRV CRD1 INTR1 C D E F G A B
	


	UPD.F1
	UPDate and return the Frequency of the root note
	

	Code
	« UPD.N1 3 GET IF 1 FS? THEN R * ELSE IF 2 FS? THEN R / END END »
	

	Stack
take/give
	1: "Chord1"
	2: "Chord1"

1: frequency

	Result
	Uses argument to creat/update CRD1, INTRV, flags 1 to 12, and return the frequency of the root note. 
	

	Flags
	1 2 3 4 5 6 7 8 9 10 11 12
	

	Names
required
	UPD.N1 UPD1 UPD2 UPD3 INTRV CRD1 INTR1 C D E F G A B
	


	UPD1
	UPDate program 1
	

	Code
	« 'CRD1' STO 1 12 FOR f f CF NEXT IF CRD1 DUP "#" POS THEN 1 SF ELSE IF DUP "b" POS THEN 2 SF END END »
	

	Stack
take/give
	1: "Chord1"
	1: "Chord1"

	Result
	Stores argument in CRD1, clear flags 1 to 12, update flag 1, 2 and then return the argument unchanged.
(UPD1 UPD2 UPD3 are always used together)
	

	Flags
	1 2
	


	UPD2
	UPDate program 2
	

	Code
	« IF DUP "7" POS THEN 12 SF IF DUP "M" POS THEN 7 SF ELSE IF DUP "m" POS THEN 9 SF ELSE IF DUP "h" POS THEN 10 SF ELSE IF DUP "˚" POS THEN 11 SF ELSE 8 SF END END END END END »
	

	Stack
take/give
	1: "Chord1"
	1: "Chord1"

	Result
	UPD2 takes the argument left by UPD1, check the chord quality to make flags 7-12 right, then return the argument it took from UPD1 untouched.
(UPD1 UPD2 UPD3 are always used together)
	

	Flags
	7 8 9 10 11 12
	


	UPD3
	UPDate program 3
	

	Code
	« IF 12 FC? THEN IF DUP "M" POS THEN 3 SF ELSE IF DUP "m" POS THEN 4 SF ELSE IF DUP "+" POS THEN 5 SF ELSE 6 SF END END END END INTRV »
	

	Stack
take/give
	1: "Chord1"
	1: "Chord1"

	Result
	Takes argument (left by UPD2), check the argument's quality to set flags 3-6 (triads) right, set new INTR1, then fianlly return the argument to stack.

(result of running UPD1, UPD2 & UPD3 is a updated CRD1, INTR1, and flags 1 to 12 using the argument (on stack one), then return the argument unchanged to stack one. )
	

	Flags
	3 4 5 6
	

	Names
required
	INTRV
	


	INTRV
	get INTeRVals sequence for the current chord
	

	Code
	« IF 7 FS? THEN { 0 4 7 11 } ELSE IF 8 FS? THEN { 0 4 7 10 } ELSE IF 9 FS? THEN { 0 3 7 10 } ELSE IF 10 FS? THEN { 0 3 6 10 } ELSE IF 11 FS? THEN { 0 3 6 9 } ELSE IF 3 FS? THEN { 0 4 7 12 } ELSE IF 4 FS? THEN { 0 3 7 12 } ELSE IF 5 FS? THEN { 0 4 8 12 } ELSE IF 6 FS? THEN { 0 3 6 12 } END END END END END END END END END 'INTR1' STO »
	

	Stack
take/give
	none
	? (should be none)

	Result
	Update INTR1.

(This is done by checking flags 3 to 11.)
	

	Names
required
	INTR1
	


	PLAY1
	PLAY a chord
	

	Code
	« UPD.F1 LP1 »
	

	Logic
	Takes argument "Chord1" then uses subprogram UPD.F1 to return the frequency for the first note of the chord, then LP1 takes the argument to play the chord.
	

	Stack
take/give
	"Chord1"
	"Chord1"

	Result
	Plays argument in root and closed position three times in arpeggio style. Update CRD1, INTR1, flags 1 to 12 and return argument unchanged to stack one.
	

	Flags
	1 2 3 4 5 6 7 8 9 10 11 12
	

	Names
required
	UPD.F1 LP1 UPD1 UPD2 UPD3 INTRV CRD1 INTRV R DUR  C D E F G A B
	


	LP1
	put LooPing to work for PLAY1
	

	Code
	«  f1 « 1 3 START IF 12 FS? THEN 5 ELSE 4 END 1 SWAP FOR t R INTR1 t GET ^ f1 * DUR BEEP NEXT NEXT » »
	

	Stack
take/give
	1: frequency (of the root)
	none

	Result
	Uses the argument and play the current chord in arpeggio style 3 times.
	

	Names
required
	INTRV R DUR
	


	PLAY2
	PLAY a chord
	

	Code
	« UPD.F1 LP2 »
	

	Logic
	Uses subprogram UPD.F1 then LP2 to complete function.
	

	Stack
take/give
	"Chord1"
	"Chord1"

	Result
	Plays the chord in stack one in root and closed position in arpeggio style for 5 times. Each time an octave higher. Update CRD1, INTR1, and flags 1 to 12. Argument is returned unchanged.
	

	Flags
	1 2 3 4 5 6 7 8 9 10 11 12
	

	Names
required
	UPD.F1 LP2 UPD1 UPD2 UPD3 INTRV CRD1 INTRV R DUR C D E F G A B
	


	LP2
	put LooPing to work for PLAY2
	

	Code
	«  f1 « 0 5 FOR oc IF 12 FS? THEN 4 ELSE 3 END 1 SWAP FOR t R INTR1 t GET ^ 2 oc ^ * f1 * DUR BEEP NEXT NEXT » »
	

	Stack
take/give
	1: frequency (of the root)
	none

	Result
	Uses the argument and play the current chord in root and closed position in arpeggio style for 5 times. Each time an octave higher.
	

	Flags
	
	

	Names
required
	INTRV R DUR
	


	PLAY3
	PLAY a chord
	

	Code
	« UPD.F1 LP3 »
	

	Logic
	Uses subprogram UPD.F1 then LP3 to complete function.
	

	Stack
take/give
	"Chord1"
	"Chord1"

	Result
	Plays the chord in stack one in root and closed position in arpeggio style for 5 times. Each time an octave lower. Update CRD1, INTR1, and flags 1 to 12. Argument is returned unchanged.
	

	Flags
	1 2 3 4 5 6 7 8 9 10 11 12
	

	Names
required
	UPD.F1 LP2 UPD1 UPD2 UPD3 INTRV CRD1 INTRV R DUR C D E F G A B
	


	LP3
	put LooPing to work for PLAY3
	

	
	«  f1 « 5 0 FOR oc IF 12 FS? THEN 4 ELSE 3 END 1 FOR t R INTR1 t GET ^ 2 oc ^ * f1 * DUR BEEP -1 STEP -1 STEP » »
	

	Stack
take/give
	1: frequency (of the root)
	none

	Result
	Uses the argument and play the current chord in root and closed position in arpeggio style for 5 times. Each time an octave higher.
	

	Names
required
	INTRV R DUR
	


	CHORD
	generate a CHORD
	

	Code
	« NOTE BCEF « RAND IF 0.4999 > THEN Q3RD ELSE Q7TH END » EVAL + »
	

	Logic
	Using NOTE BCEF to generate a chord, then randomly choose Q3RD or Q7TH to add the quality. Results are all the possible triads and 7th chords except enharmonic chords with root on B#, E#, Cb, or Fb. 
	

	Stack
take/give
	none
	1: "Chord"

	Result
	Returns an arbitrary "Chord" to stack one.
The returned chord will not have roots on B#, E#, Cb, or Fb.
	

	Names
required
	NOTE BCEF Q3RD Q7TH
	


	NOTE
	generate music NOTEs 
	

	Code
	« « { "A" "B" "C" "D" "E" "F" "G" } RAND 100 * FLOOR 7 MOD 1 + GET » EVAL « { "#" " " "b" } RAND 100 * FLOOR 3 MOD 1 + GET » EVAL + »
	

	Stack
take/give
	none
	1: "Note"

	Result
	Returns a string such as "A#", "A ", "Ab" ... etc.
	


	BCEF
	checks notes of the letter name B C E or F
	

	
	«  { "B#" "E#" "Cb" "Fb" }  A B « 0 1 4 FOR j B j GET A SAME OR NEXT IF THEN A NUM CHR " " + ELSE A END » »
	

	Stack
take/give
	1: "Note1"
	1: "Note2"

	Result
	Takes "Note1" and returns "Note2".
If "Note1" is "B#", "E#", "Cb", or "Fb", then BCEF returns the note without accidental, e.g. "B ", otherwise BCEF leave the note untouched.
The purpose of BCEF is to avoid awkward enharmonic roots tha may entail double sharp or flats when spelled.
	


	Q3RD
	Quality for triads (consisting 3RDs).
	

	Code
	« { "M" "m" "h" "°" } RAND 100 * FLOOR 4 MOD 1 + GET " " + »
	

	Stack
take/give
	none
	1: "ChordQuality"

	Result
	Returns one of the following: "M ", "m ", "+ ", or "˚ ".
Q3RD generate a quality for a triad randomly.
	


	Q7TH
	Quality for 7TH chord
	

	Code
	« { "M" " " "m" "h" "°" } RAND 100 * FLOOR 5 MOD 1 + GET "7" + »
	

	Stack
take/give
	none
	1: "ChordQuality"

	Result
	Returns one of the following: "M7" "7" "m7" "h7" "˚7"
Q7TH generate a quality for 7th chord randomly.
	


Variables:
	
	Content/description
	

	CRD1
	A string (denoting a chord). e.g. "A M " for A major triad.

Description: CRD1 is the most recent chord (current chord) spelled or played. CRD1 is created by UPD1. Programs changing the value of CRD1 are: SPELL PLAY1 PLAY2 PLAY3 UPD.F1 UPD.M1 UPD.N1 UPD1.
	

	INTR1
	A list consisting four integers. e.g. { 0 4 7 11 }.
Description: INTR1 is the subsequent intervals (measured in semi-steps) of the current chord in list format.
e.g.: { 0 4 7 11 } for a major 7th chord. 0 is the semi-steps from the 1st note to 1st note, 4 is the semi-steps needed from the 1st note to 2nd note etc. If the current chord is a triad (3 notes), then the last value (12) is the repeated root. e.g.: { 0 4 7 12 } for a major triad.

INTR1 is created and updated by INTRV.  
	

	R
	1.05946309436
description: R is the ratio of adjacent pitchs in equal tempered tuning system. R = 2^(1/12). You may change R for interesting effects.
	

	DUR
	A positive real number.
Description: DUR is the duration in seconds of each note to be played. You can store a desired value. Default is 0.1.
	


Constants:
	C
	{ 36 { "C" "E" "G" "B" } 65.406 }
description: 36 is the MIDI code for C1, { "C" "E" "G" "B" } is the letter names for chords with root C. 65.406 is the frequency of C1.
	

	D
	{ 38 { "D" "F" "A" "C" } 73.416 }
	

	E
	{ 40 { "E" "G" "B" "D" } 82.407 }
	

	F
	{ 41 { "F" "A" "C" "E" } 87.307 }
	

	G
	{ 43 { "G" "B" "D" "F" } 97.999 }
	

	A
	{ 45 { "A" "C" "E" "G" } 110 }
	

	B
	{ 47 { "B" "D" "F" "A" } 123.471 }
	

	
	
	


� In HP-28S' nomenclature, Name includes: Procedure (Program, Algebraic), Data (Real Number, String, List, Matrix, Vector, Binary Integer), or other Names.





